1. **(Addition of arbitrary precision numbers.)** Consider the following representation of base-10 numbers: the number with (familiar, base-10) digit sequence $d_k d_{k-1} \ldots d_1$ is represented as the list $(d_k \ldots d_1)$ containing the number's digits (so the most significant digit is given first). Thus, the number 371 is expressed as the list `(3 7 1)`. Define a function `apa-add` which takes two numbers in this representation and returns their sum (in the same representation). (`apa` stands for *arbitrary precision arithmetic*.)

   (Hint: It may simplify your thinking about the problem if you work with the *reverse* of this representation. Note that if n is a (positive, whole) number `(modulo n 10)` returns the least-significant digit (the "ones" digit) and `(quotient n 10)` returns the larger digits.)

   We collect two different solutions below:

```
(define (apa-add-reverse n1 n2)
  (define (rem-digits digs)
    (if (null? digs) '() (cdr digs)))
  (define (apa-add-helper p1 p2 carry)
    (if (and (null? p1)
             (null? p2)
             (= 0 carry))
        '()
        (let* ((d1 (if (null? p1) 0 (car p1)))
               (d2 (if (null? p2) 0 (car p2)))
               (total (+ carry d1 d2)))
          (cons
           (modulo total 10) (apa-add-helper
                              (rem-digits p1)
                              (rem-digits p2)
                              (quotient total 10))))))
  (apa-add-helper n1 n2 0))

(define (apa-add num1 num2)
  (reverse (apa-add-reverse (reverse num1) (reverse num2))))
```

   You'll notice that one hassle is dealing appropriately with the base cases, or cases when one number is longer than the other. The second solution handles this differently:

```
(define (normalize n k)
  (define (pad n to)
    (if (<= to 0) n
        (cons 0 (pad n (- to 1)))))
  (pad n (- k (length n))))

(define (apa-add n1 n2)
  (define (apa-add-aux n1 n2 carryIn)
    (cond ((and (null? n1) (null? n2)) (if (= carryIn 0)
                                           '() (list carryIn)))
          (else (let* ((s (+ (car n1) (car n2) carryIn))
                       (carryOut (quotient s 10))
```

```
                    (digit    (modulo s 10)))
               (cons digit (apa-add-aux (cdr n1)
                                        (cdr n2)
                                        carryOut))))))
  (let ((lr  (max (length n1) (length n2))))
    (reverse (apa-add-aux (reverse (normalize n1 lr))
                          (reverse (normalize n2 lr)) 0))
  ))
```

2. With the same representation as above, define a function `d-multiply`, which multiplies a number in this list representation by a digit (that is, a number in the set $\{0, 1 \ldots, 9\}$). Thus, for example, `(d-multiply '(1 2 3) 3)` should return `(3 6 9)`.

```
(define (apa-mult-digit n1 d)
  (define (apa-mult-aux n1 d carryIn)
    (cond ((null? n1) (if (= carryIn 0) '() (list carryIn)))
          (else (let* ((m         (+ (* (car n1) d) carryIn))
                       (carryOut  (quotient m 10))
                       (digit     (modulo m 10)))
                  (cons digit (apa-mult-aux (cdr n1)
                                            d
                                            carryOut))))))
  (reverse (apa-mult-aux (reverse n1) d 0)))
```

3. **(Multiplication of arbitrary precision numbers.)** Note that it is easy to multiply a number by 10 in this representation. Using this fact, and the "multiplication by a digit" definition above, give a recursive procedure to multiply two numbers together.

   (Hint: Multiplication of $d = d_n d_{n-1} \ldots d_1$ by $e = e_m e_{m-1} \ldots e_1$ can be carried out by the rule

   $$de = d * e_1 + 10 * (d * e_m e_{m-1} \ldots e_2).)$$

```
(define (apa-mult n1 n2)
  (define (times10 n)
    (cond ((null? n) (list 0))
          (else (cons (car n) (times10 (cdr n))))))
  (define (apa-mult-aux n1 n2)
    (cond ((null? n2) (list 0))
          (else  (let ((fd (car n2))
                       (rd (cdr n2)))
                   (apa-add (apa-mult-digit n1 fd)
                            (times10 (apa-mult-aux n1 rd)))))))
  (apa-mult-aux n1 (reverse n2)))
```

4. **(Surgery on Binary Search Trees.)** Suppose that T is a binary search tree as shown on the left hand side of Figure 1. As T is a binary search tree, we must have $b < a$; furthermore, all elements in T_1 are smaller than b, all elements in T_2 lie in the range (a, b), and all elements of T_3 are larger than a. Consider now the operation of *right rotation* of T which re-arranges the root and subtrees as shown. It is easy to check that while this "right rotation" changes the structure of the tree, the resulting tree still satisfies the binary search tree property. (The same can be said of "left rotation.")



Figure 1: *Right rotation* and *left rotation* of a binary search tree.

The *depth* of an element in a binary tree is the length of the path from the root to the element. It is desirable, for binary search trees, to arrange them so that elements have small depth. (Recall that the *depth* of the tree is the length of the longest path in the tree; equivalently, it is the depth of the "deepest" element.)

Thus, if the depth of T_1 is more than one larger than the depth T_3, a right rotation makes progress toward balancing the tree. Likewise, if the depth of T_3 is more than one more than the depth of T_1, a left rotation seems to be just the thing to improve the tree.

Write a function called tree-repair which takes, as an argument, a binary search tree. tree-repair should use left and right rotations to try to balance the tree T, as follows:

- Recursively repair each of T's two subtrees. (So, you must build a new tree whose two subtrees are obtained by repairing the two subtrees of T.)
- Once the subtrees have been repaired, examine T. If $\text{depth}(T_1) > \text{depth}(T_3) + 1$ return the result of rotating T to the right. If, on the other hand, $\text{depth}(T_3) > \text{depth}(T_1) + 1$ return the result of rotating T to the left.

It would probably be a good idea to start by writing functions that carry out left and right rotations.

```
(define (depth T)
  (if (null? T)
      -1
      (+ 1 (max (depth (left T))
                (depth (right T))))))


(define (improve T)
  ; Define right and left rotations
  (define (rotate-right T)
    (let ((T1 (left (left T)))
          (T2 (right (left T)))
          (T3 (right T)))
      (make-tree (value (left T))
                 T1
                 (make-tree (value T) T2 T3))))
  (define (rotate-left T)
    (let ((T1 (left T))
          (T2 (left (right T)))
          (T3 (right (right T))))
      (make-tree (value (right T))
                 (make-tree (value T) T1 T2)
                 T3)))
```

```
  ; The logic to determine if a rotation
  ; would improve things
  (define (right-improvement? T)
    (and (not (null? T))
         (not (null? (left T)))
         (> (depth (left (left T)))
            (+ 1 (depth (right T))))))
  (define (left-improvement? T)
    (and (not (null? T))
         (not (null? (right T)))
         (> (depth (right (right T)))
            (+ 1 (depth (left T))))))

  ; Finally, the code to make a single improvement
  (define (improve-once S)
    (cond ((left-improvement? S) (rotate-left S))
          ((right-improvement? S) (rotate-right S))
          (else S)))

  ; The body: improve each child, and try to
  ; improve the root
  (if (null? T) T
      (improve-once
       (make-tree (value T)
                  (improve (left T))
                  (improve (right T))))))
```

5. **(Heapsort.)** Write a sorting procedure which, given a list of numbers, returns a list of the same numbers in sorted order. Your procedure should do the following:

   - Add all of the elements of the initial list into a heap; be careful to arrange your inserts so that the heap remains balanced. (For this purpose, use the heuristic we discussed in class: always insert into the left child and exchange the order of the children.)
   - Repeatedly *extract-min* (extract the minimum element) from the heap, and return the (sorted) list of elements gathered in this way.

   First, a purely functional heap implementation.

```
(define (hsort elements)
  ; Tree helper functions
  (define (make-tree value left right)
    (list value left right))
  (define (value T) (car T))
  (define (left T) (cadr T))
  (define (right T) (caddr T))

  ; Heap management functions
  (define (combine Ha Hb)
    (cond ((null? Ha) Hb)
          ((null? Hb) Ha)
          ((< (value Hb) (value Ha)) (combine Hb Ha))
```

```
          (else (make-tree (value Ha)
                           (combine (left Ha) (right Ha))
                           Hb))))
  (define (heap-insert x H)
    (cond ((null? H) (make-tree x '() '()))
          ((< x (value H))
           (make-tree x
                      (heap-insert (value H) (right H))
                      (left H)))
          (else
           (make-tree (value H)
                      (heap-insert x (right H))
                      (left H)))))

  ; Finally, a function that inserts an entire list into a heap.
  (define (heap-insert-list elements H)
    (if (null? elements)
        H
        (heap-insert-list (cdr elements)
                          (heap-insert (car elements) H))))
  (define (heap-min H) (value H))
  (define (heap-remove-min H) (combine (left H) (right H)))

  ; Finally, a function that extracts an entire sorted
  ; list from a heap.
  (define (heap-extract-sorted H)
    (if (empty? H) '()
        (cons (heap-min H)
              (heap-extract-sorted (heap-remove-min H)))))

  ; The main call.
  (heap-extract-sorted (heap-insert-list elements '())))
```

Here is a version that implements the heap as a (destructive) object. Internal heap operations are still done functionally.

```
(define (make-heap)
  (let ((theHeap '()))

    (define (make-tree value left right)
      (list value left right))
    (define (value T) (car T))
    (define (left T) (cadr T))
    (define (right T) (caddr T))

    (define (combine Ha Hb)
      (cond ((null? Ha) Hb)
            ((null? Hb) Ha)
            ((< (value Hb) (value Ha)) (combine Hb Ha))
            (else (make-tree (value Ha)
                             (combine (left Ha) (right Ha))
                             Hb))))
```

```
    (define (heap-insert x workHeap)
      (cond ((null? workHeap) (make-tree x '() '()))
            ((< x (value workHeap))
             (make-tree x
                        (heap-insert (value workHeap)
                                     (right workHeap))
                        (left workHeap)))
            (else
             (make-tree (value workHeap)
                        (heap-insert x (right workHeap))
                        (left workHeap)))))

    (define (empty?) (null? theHeap))
    (define (insert x)
      (set! theHeap (heap-insert x theHeap)))
    (define (extract-min)
      (let ((smallest (value theHeap)))
        (begin
          (set! theHeap (combine (left theHeap)
                                 (right theHeap)))
          smallest)))
    (define (insert-list elements)
      (if (null? elements)
          #t
          (begin (insert (car elements))
                 (insert-list (cdr elements)))))
    (define (extract-sorted)
      (if (empty?) '()
          (cons (extract-min)
                (extract-sorted))))

    (lambda (method)
      (cond ((eq? method 'empty) empty?)
            ((eq? method 'insert) insert)
            ((eq? method 'insert-list) insert-list)
            ((eq? method 'extract-min) extract-min)
            ((eq? method 'extract-sorted) extract-sorted)
            ))
    )
  )

(define (heapsort elements)
  (let ((heap (make-heap)))
    (begin
      ((heap 'insert-list) elements)
      ((heap 'extract-sorted)))))
```