

Laboratory Assignment 7

Objectives

- More work with lists
- Using the map function

Activities

1. Use tail-recursion to write a SCHEME function, named `list-sum` which takes a list of numbers as a parameter and computes their sum. *Note: this function differs slightly from the one you must write to complete the problem set this week. Be sure to follow the directions in the problem set precisely when preparing your submission.*

Solution:

```
(define (list-sum l)
  (define (list-sum-aux lst sum)
    (if (null? lst)
        sum
        (list-sum-aux (cdr lst) (+ (car lst) sum))))
  (list-sum-aux l 0))
```

2. The mean, or average, of a set of n numbers is the sum of all the numbers divided by the quantity n . Write a SCHEME function, named `mean`, which takes a list of numbers as a parameter and returns the mean of the numbers in the list.

Solution:

```
(define (list-mean l)
  (/ (list-sum l) (length l)))
```

3. The median value of a set of n numbers is the value that separates the half of higher values from the half of lower values in the set. The median can be found by arranging the values in the set in order and choosing the “middle” value. If there are an even number of values in the set, the median is described as the mean of the two middle values. Write a SCHEME function, named `list-median`, that takes a list of numbers as a parameter and returns the median value in the list.

Solution:

```
;uses qsort and partition from the lecture slides
(define (list-median l)
  ;lists are defined here starting with index 0
  (define (list-at l i)
```

```

    (if (equal? i 0)
        (car l)
        (list-at (cdr l) (- i 1))))
(define (is-even x) (= (modulo x 2) 0))
(let ((sorted (qsort l)))
    (if (is-even (length l))
        (/ (+ (list-at sorted (floor (/ (length l) 2)))
              (list-at sorted (- (floor (/ (length l) 2)) 1)))
           2)
        (list-at sorted (floor (/ (length l) 2)))))

```

4. In probability and statistics, variance, denoted by the symbol sigma squared, σ^2 , is a measure of how “spread out” the numbers in a set are. Write a SCHEME function, named `list-variance`, which takes a list of numbers as a parameter and uses the built-in `map` function and the `list-sum` function in question 1 to compute the variance of the numbers in the list. It can be computed using the following function:

$$\sigma^2 = \frac{1}{n} \sum_{i=1}^n (x_i - \bar{x})^2$$

where \bar{x} is the mean of the numbers in the set and x_i is the i^{th} number in the set.

Solution:

```

(define (list-variance l)
  (define (square x) (* x x))
  (let* ((average (list-mean l))
        (v-list (map (lambda (x) (square (- x average))) l)))
    (/ (list-sum v-list) (length l))))

```

5. Also in probability and statistics, the standard deviation, denoted by the symbol sigma, σ , of a set of numbers is the square-root of the variance of the set. Write a SCHEME function, named `list-stddev`, which takes a list of values as a parameter and uses the `list-variance` function in question 4 to compute their standard deviation.

Solution:

```

(define (list-stddev l)
  (sqrt (list-variance l)))

```